
Physics 651: Assignment 1
(to be submitted by Tuesday, September 10, 2024)

I invite you to attempt Assignment 1 and to turn in your work for Questions 2, 4, 6, 8, and 10. Your sub-
mission should take the form of a single Wolfram Notebook, sent as an attachment to kbeach@olemiss.
edu. Please follow the naming convention Phys651-A1-webid.nb, and be sure to include the subject line
Phys651-Fall2024-webid Assignment 1 Submission.

1. You can think of the finite field 𝔽𝑛 as the ring of integers {0, 1, 2,… , 𝑛 − 1}, defined modulo 𝑛.

(a) Generate lists of the elements in 𝔽5 inMathematica using Range and Table. Both of these commands
produce a comma-separated collection of integer values, enclosed in matching (curly) braces. Note
that pressing enter or return on your keyboard will produce just a bare line-feed/carriage-return. To
getMathematica to interpret and execute a line of code, you will have to type shift + return .

Range[0, 4]
Table[i, {i, 0, 4}]

(b) Next, generate the set 𝐴 = {1∕(1 + 𝑘2) ∶ 𝑘 ∈ 𝔽5}. Here are four ways to do so:

A = 1/(1 + Range[0, 4]^2)
A = Map[Function[1/(1 + #^2)], Range[0, 4]]
A = Map[1/(1 + #^2) &, Range[0, 4]]
A = Array[1/(1 + (# - 1)^2) &, 5]

In the second and third lines above, Map applies the anonymous function 𝑘 → 1∕(1+𝑘2)—expressed
in code as 1/(1+#^2) and wrapped in a Function call or terminated with the & symbol—to each ele-
ment of the list.
Write your own code to compute 𝐴, based on a call to Table.

(c) Determine the addition and multiplication tables for 𝔽3. Compute all the elements by hand.
(d) Verify that your calculation agrees with the two tables produced algorithmically by the following

Mathematica code:

Table[Table[Mod[i+j, 3], {i, 0, 2}], {j, 0, 2}] // Grid
Table[Table[Mod[i*j, 3], {i, 0, 2}], {j, 0, 2}] // Grid

The double slash is the function postfix operator, meaning that expr // f is equivalent to f[expr],
and Grid arranges the elements in a 3× 3 grid. (Other formatting options include // MatrixForm and
// TraditionalForm.)
As a convenience, the Table function supports a simplified syntax for nested tables.

Table[Mod[i+j, 3], {i, 0, 2}, {j, 0, 2}] // Grid
Table[Mod[i*j, 3], {i, 0, 2}, {j, 0, 2}] // Grid

Also try this with the // Grid specification removed, which lets you see the unformatted result. Ob-
serve that the output is a nested list of lists. This is the data structure that Mathematica uses for
matrices and multidimensional arrays.
Alternatively, one can map an anonymous function that handles the modulo arithmetic onto the
outer product (Outer) of the lists produced by Range.

Outer[Function[Mod[#1+#2, 3]], Range[0, 2], Range[0, 2]] // Grid[#, Frame -> All] &
Outer[Mod[#1*#2, 3] &, Range[0, 2], Range[0, 2]] // Grid[#, Frame -> All] &

2. (a) Take inspiration from question 1d to produce the formatted addition and multiplication tables of 𝔽𝑛
algorithmically. Write the code for general 𝑛, but show a test result with the specific value n = 12.

(b) Consider a triply nested Table structure representing all possible sums of three numbers in 𝔽5. We
can then confirm that the set of all resulting values covers 𝔽5.

T = Table[Mod[i+j+k, 5], {i, 0, 4}, {j, 0, 4}, {k, 0, 4}]
Flatten[T]
Union[Flatten[T]]
DeleteDuplicates[Flatten[T]] (* This is equivalent to the previous line *)
Range[0,4] == %

Explain what is going on in the last four lines of this code snippet. What does Flatten do? What
purpose do Union and DeleteDuplicates serve? What is the meaning of the % symbol?
Finally, rewrite the code using Outer and Range.

(c) Create a quadruply nested list structure representing all possible products of four numbers in 𝔽10.
Verify that the set of values covers 𝔽10.

3. Mathematica offers a variety of input methods for special characters and symbols. Most characters have a
full name (\[FullName]) and an associated alias ( esc alias esc ) for faster input. For example, ∞, 𝜋,
and 𝑒 are produced by \[Infinity] \[Pi], and \[ExponentialE] but also by esc inf esc , esc pi esc ,
and esc ee esc . Some characters also have a standalone name that doesn’t require an escape sequence:
Infinity and E can be used unadorned. Run the following code to establish membership of various num-
bers and expressions in the integers (ℤ), reals (ℝ), and rationals (ℚ).

Element[-256, Integers]
Element[11/32, Integers]
Element[\[Infinity], Reals]
0 esc el esc NonNegativeReals
Element[\[Pi], Reals]
esc ee esc esc el esc Rationals

Element[-2/3, Rationals]
2/3 \[Element] Reals \[And] 2/3 \[Element] Rationals
Element[52, Reals] esc and esc Element[52, Integers]
(* The Sinc function has a well-defined real-valued limit at zero. *)
Element[Limit[Sin[x]/x, x -> 0],Reals]
(* Every individual element of this summation is rational, but the total sum is real. *)
Apply[And, Table[1/n^2 \[Element] Rationals, {n, 1, 100}]]
Sum[1/n^2, {n, 1, Infinity}] esc el esc Reals

4. As aMathematica coding exercise, show that the infinite sum
∑∞

𝑛=0(−1)
𝑛∕𝑛! = 1∕𝑒 is real, whereas each

of the first 101 partial sums,

0∑

𝑛=0

(−1)𝑛

𝑛! ,
1∑

𝑛=1

(−1)𝑛

𝑛! ,
2∑

𝑛=0

(−1)𝑛

𝑛! , … ,
100∑

𝑛=0

(−1)𝑛

𝑛! ,

is individually rational. You may find it helpful first to investigate your options for summation and
Boolean manipulation with queries to ?Sum, ?And, and ?Equal.

5. Mathematica supports complex numbers via an imaginary 𝑖 =
√
−1, represented by \[ImaginaryI],

esc ii esc , or just I. Hence, 1 + 2I and 2 - 3*I are valid numbers in ℂ (Complexes). Multiplication
is inferred, so products can be expressed as either a space b or a*b. In the following code, note the dif-
ference between the assignment operator (=) and the test-for-equality operator (==). Further observe that
logical and has both an operator (&&) and function form (And). Take care to distinguish the semantics of a
Symbol and String. The latter supports concatenation (via the function StringJoin or operator <>).

(* What follows are four ways to define the same complex number *)
a1 = Complex[1,2]
a2 = 1+2\[ImaginaryI]
a3 = 1+2 esc ii esc
a = 1+2I
(* Here we verify that they are equal*)
EqualTo[a1][a]
a1 == a && a2 == a
a1 == a esc and esc a2 == a esc and esc a3 == a
Apply[And,Table[Symbol["a" <> ToString[i]] == a , {i, 1, 3}]]
(* The complex numbers are closed under addition, subtraction, multiplication, and

division*)
b = 2 + 3I
z1 = a+b
z2 = a-b
z3 = a b
z4 = a/b
z1 esc el esc Complexes
z2 esc el esc Complexes
Apply[And,Table[Symbol["z" <> ToString[i]] \[Element] Complexes , {i, 1, 4}]]
Norm[z1]
N[Norm[z1]]
Arg[z1]
N[%, 20]
Norm[z2]
Arg[z2]
For[k = 1, k < 5, ++k, zklabel = "z" <> ToString[k]; zk = Symbol[zklabel];
Print[zklabel, " = ", zk, ": ", "\nnorm(", zklabel, ") = ", Norm[zk],
",\narg(", zklabel, ") = ", Arg[zk]]]

Explicit numerical evaluation is carried out with the function N, which takes an optional argument that
specifies the precision (the default is 8 decimal digits). The For loop uses aC-like initialize–test–increment–
body syntax. Print outputs text and mathematical expressions to the screen; it is aware of standard string
escape sequences, such as \t (tab) and \n (newline).

6. Write a program that defines complex numbers 𝑧𝑘 = cos(𝜋𝑘∕5)+𝑖 sin(𝜋𝑘∕5) for 𝑘 = 0, 1,… , 9 and for each
𝑘 verifies that |𝑧𝑘| = 1, Arg 𝑧𝑘 = 𝜋𝑘∕5 (mod 2𝜋), and exp(𝑖Arg 𝑧𝑘) ≐ 𝑧𝑘. Where ≐ is indicated, establish
that the expressions on the left- and right-hand side agree to within 12 digits; otherwise, demand exact
(symbolic) equality.

7. The value 𝜙 = (1 +
√
5)∕2 can be computed with the expression (1+Sqrt[5])/2 or directly invoked with

the special named constant GoldenRatio. 𝜙 is unique in that it has an infinite continued fraction repre-
sentation in which all the terms are 1:

𝜙 = 1 + 1

1 + 1

1+ 1
1+⋯

Mathematica has some useful helper functions for continued fractions. ContinuedFraction converts a
number or mathematical expression into a (truncated) list of its coefficients in the continued fraction
representation. Conversely, FromContinuedFraction takes the finite list of coefficients and evaluates the
corresponding number as a partial continued fraction. It’s instructive to observe how the partial continued
fractions converge to 𝜙 as the length of the coefficient list grows. Take time to be sure that you understand
the ReplaceAll (/.) trick that allows us to compute the terms explicitly.

ContinuedFraction[GoldenRatio, 1]
FromContinuedFraction[%]
1
ContinuedFraction[GoldenRatio, 2]
FromContinuedFraction[%]
1+1
ContinuedFraction[GoldenRatio, 3]
FromContinuedFraction[%]
1+1/(1+1)
ContinuedFraction[GoldenRatio, 4]
FromContinuedFraction[%]
1+1/(1+1/(1+1))
ContinuedFraction[GoldenRatio, 5]
FromContinuedFraction[%]
1+1/(1+1/(1+1/(1+1)))
expr = 1 + w; Do[expr = expr /. w -> 1/(1 + w); Print[N[expr /. w -> 0, 12]], 20]
ListLinePlot[Table[{k, FromContinuedFraction[ContinuedFraction[GoldenRatio, k]]}, {k, 1,

10}], PlotMarkers -> Automatic, PlotRange -> {{0, 11}, {0.5, 2.5}}, Frame -> True]

8. Now consider the sequence of partial continued fractions,

𝜋1 = 3, 𝜋2 = 3 + 1
7 , 𝜋3 = 3 + 1

7 + 1
15

, 𝜋4 = 3 + 1

7 + 1

15+ 1
1

, 𝜋5 = 3 + 1

7 + 1

15+ 1

1+ 1
292

, …

Each 𝜋𝑘 is the continued fraction that has been truncated at order 𝑘. These achieve the asymptotic value
lim𝑘→∞ 𝜋𝑘 = 𝜋; the sequence approaches its limit from below. Investigate the convergence by making
a table of the first 15 values (𝜋1, 𝜋2, …, 𝜋15). Then make a plot (with a logarithmic vertical axis, using
ListLogPlot) of 𝜋 − 𝜋𝑘 for 𝑘 ranging from 1 to 50.

9. Mathematica supports various Function styles and other syntactic sugar (UseShorthandNotations). The
hash (#) represents the single argument passed to a unary function; #1, #2, … represent the various argu-
ments passed to amultivariable function. The ampersand (&) is the terminator for an anonymous function
definition. The operators /@, @@, and @@@ are stand-ins for Map, Apply, and MapApply. List-element selection
and manipulation (ElementsOfLists) can be carried out with the functions Take, Drop, Part, and Span but
also with a double-square-bracket indexing-and-slicing notation ([[;;]]).

(* c1 is a list of complex numbers; c2 and c3 are identity transformations of c1 *)
c1 = 1 + Range[5] I
c2 = Map[Norm[#] Exp[I Arg[#]] &, c1]
c3 = Norm[#] Exp[I Arg[#]] & /@ c1
N[c1]
N[c2]
c3 // N
Table[Simplify[c1[[i]] == c2[[i]]], {i, 0, 5}]
Simplify[MapThread[#1==#2&,{c1,c2}]]
Simplify[Thread[c1==c2]]
Simplify[Equal @@@ Table[{c1[[i]], c2[[i]]}, {i, 1, Length[c1]}]]
(* Plot spirals in the complex plane *)
F[a_, n_] := (1 + a Range[0, n]/(n/2)) Exp[(\[Pi]/(2 n/5)) I Range[0, n]];
ComplexListPlot[{F[2, 200], F[3, 200], F[4, 200]}, PlotLegends -> {"a=2", "a=3", "a=4"}]
(* There are several ways to index and slice lists *)
u = Fibonacci /@ Range[15]
u[[1]]
u[[-1]] (* This is a python-style syntax that selects the last element of the list *)
u[[9]]
Part[u,9]
u[[3 ;; 7]]
Take[u, {3, 7}]
u[[;; 5]]
Take[u, 5]
u[[-5 ;;]]
Take[u, -5]

10. The famous Fibonacci sequence (𝑓𝑛) = (𝑓1, 𝑓2, 𝑓3,…) = (1, 1, 2, 3, 5, 8, 13, 21, 34, 55,…) is defined recur-
sively by 𝑓1 = 𝑓2 = 1 and 𝑓𝑛+1 = 𝑓𝑛 + 𝑓𝑛−1. These statements both evaluate to true:

Fibonacci[1] == Fibonacci[2]
And @@ ((Fibonacci[# + 1] == Fibonacci[#] + Fibonacci[# - 1]) & /@ Range[1, 100])

Consider a sequence transformation (𝑓𝑛 → 𝑔𝑛 = (𝑓3𝑛−2𝑓3𝑛−1𝑓3𝑛)1∕3) that maps three consecutive Fi-
bonacci values into their geometric average:

(𝑔𝑛) = ((𝑓1𝑓2𝑓3)1∕3, (𝑓4𝑓5𝑓6)1∕3, (𝑓7𝑓8𝑓9)1∕3, …).

Using Mathematica, populate a list with the first 10 elements of (𝑔𝑛). As a challenge, see if you can
accomplish this with a program consisting of only one line of code. The list elements should be exact
symbolic results. Compare them to the following decimal approximation to confirm their correctness:

(𝑔𝑛) ≐ (1.25992, 4.93242, 21.0159, 88.9963, 377.001,…).

Finally, use Show to superimpose a ListLogPlot of 𝑔𝑛 and a LogPlot of the heuristic 0.29𝑒1.44𝑛.

